


UTILISER CAST3M

F. DI PAOLA

MAILLAGE



Documentation Cast3M 2011

<http://www-cast3m.cea.fr>

Cast3M est un logiciel de calcul par la méthode des éléments finis pour la mécanique des structures et des fluides. Cast3M est développé au Département de Modélisation des Systèmes et Structures (DM2S) de la Direction de l'Énergie Nucléaire du Commissariat à l'Énergie Atomique et aux Énergies Alternatives (CEA).

Le développement de Cast3M entre dans le cadre d'une activité de recherche dans le domaine de la mécanique dont le but est de définir un instrument de haut niveau, pouvant servir de support pour la conception, le dimensionnement et l'analyse de structures et de composants.

Dans cette optique, Cast3M intègre non seulement les processus de résolution (solveur) mais également les fonctions de construction du modèle (pré-processeur) et d'exploitation des résultats (post-traitement). Cast3M est un logiciel « boîte à outils » qui permet à l'utilisateur de développer des fonctions répondant à ses propres besoins.

Cast3M est notamment utilisé dans le secteur de l'énergie nucléaire, comme outil de simulation ou comme plateforme de développement d'applications spécialisées. En particulier, Cast3M est utilisé par l'Institut de Radioprotection et de Sécurité Nucléaire (IRSN) dans le cadre des analyses de sûreté des installations nucléaires françaises.



Table des matières

1	Introduction	7
2	Options générales pour la création de maillages	9
2.1	Couleur	9
2.2	Dimension de l'espace	9
2.3	Type d'éléments	9
2.4	Densité (taille de maille)	10
3	Définition de points	11
3.1	À partir des coordonnées	11
3.2	Par extraction sur des objets existant	11
4	Définition de lignes	13
4.1	À partir de points	13
4.2	Par extraction sur des objets existant	14
5	Définition de surfaces	15
5.1	À partir d'une ligne	15
5.2	À partir de deux lignes	16
5.3	À partir de quatre lignes	16
5.4	Par extraction sur des objets existant	16
6	Définition de volumes	17
6.1	Par remplissage d'une surface fermée	17
6.2	Par balayage d'une surface	17
7	Manipulation de maillages	19
7.1	Extraction de parties	19
7.2	Transformations géométriques	20
7.3	Modifications	20
7.4	Contrôle et caractéristiques	21
8	Exemple de réalisation de maillage	23
8.1	Bague avec rainure extérieure	23
8.2	Casserole simple en éléments coques	24
9	Aide mémoire	25
9.1	Opérateurs généraux	25
9.2	Définition de points	26
9.3	Définition de lignes	26
9.4	Définition de surfaces	27

9.5	Définition de volumes	28
9.6	Manipulation de maillages	28
9.7	Autres opérations	30
Index		33



Chapitre 1

Introduction

Ce document présente la démarche de création de maillages dans Cast3M ainsi que les opérateurs et procédures permettant de manipuler les objets de type MAILLAGE. Les principaux opérateurs, procédures et directives sont présentées de manière succincte et quelques exemples d'utilisation sont commentés. Le lecteur désireux de connaître toutes les possibilités offertes par un opérateur est invité à consulter la notice *ad hoc* en faisant INFO NOM_OPERATEUR dans Cast3M.

Dans la plupart des pré-processeurs des codes de calcul par éléments finis, les modèles géométriques sont créés en deux étapes : dans un premier temps, la définition d'une géométrie par un outil de CAO/DAO, puis dans un second temps la génération du maillage à partir de la géométrie créée. Dans Cast3M, un objet géométrique n'existe que sous forme discrétisée : la discrétisation du domaine en éléments s'effectue au moment de la définition de la géométrie. La démarche générale de construction d'un maillage est la suivante :

- construction des points ;
- construction des lignes à partir des points ;
- construction des surfaces à partir des lignes ;
- construction des volumes à partir des surfaces.

Ce document est articulé autour de chacune de ces étapes et les chapitres 3, 4, 5 et 6 présentent les principales opérations mises à la disposition de l'utilisateur pour créer des objets de maillage de cette manière.

Il est également possible de :

- créer un nouveau maillage par fusion, ou réunion, de plusieurs maillages existant grâce à l'opérateur ET ;
- créer un nouveau maillage en appliquant des transformations géométriques (translation, rotation, homothétie, ...) à un maillage existant ;
- d'extraire une partie des éléments d'un maillage existant.

La création ou la manipulation de maillages par ces types d'opérations sont d'une syntaxe commune à tous types de maillage et utilisent les mêmes opérateurs. Ces actions sont décrites dans le chapitre 7.

Enfin, nous recommandons quelques règles de « bonne conduite » à respecter lors de la création d'un maillage :

- Les entités géométriques étant créées uniquement sous forme discrétisée, il est important de paramétrer la densité du maillage (nombre et/ou taille d'éléments) à l'aide de variables si l'on souhaite modifier le maillage par la suite (raffinement).
- Il ne faut pas mailler plusieurs fois un objet déjà existant. Par exemple, si l'on a maillé un volume et que l'on souhaite obtenir le maillage d'une de ses faces, il ne faut pas remailler cette face mais extraire les éléments surfaciques correspondant à cette face à partir du volume (cf. 7.1).
- Il est vivement conseillé d'utiliser le moins possible la directive ELIMination (cf. 7.3).



Chapitre 2

Options générales pour la création de maillages

Les principaux paramètres de contrôle du maillage sont déclarés à l'aide de la directive OPTIon et de certains mots-clés.

2.1 Couleur

La couleur des éléments de maillage créés peut être modifiée grâce au mot-clé COULeur.

Syntaxe :

```
OPTI COUL MOT1 ;
```

où MOT1 est un mot pouvant prendre les valeurs suivantes : DEFA, BLEU, ROUG, ROSE, JAUN, VERT, TURQ, BLAN, NOIR, AZUR, ORAN, VIOL, OCEA, CYAN, OLIV, GRIS. On pourra également utiliser l'opérateur COULeur.

2.2 Dimension de l'espace

La dimension de l'espace est déclarée grâce au mot-clé DIMEnSion.

Syntaxe :

```
OPTI DIME ENTI1 ;
```

où ENTI1 est un entier valant 1, 2 ou 3. Ce paramètre est obligatoire avant toute définition de points. Il n'apparaît en principe qu'une seule fois. Toutefois, il est possible de définir des points avec OPTI DIME 2 puis de mettre OPTI DIME 3 (mais pas le contraire). Dans ce cas, la troisième coordonnée des points définis après OPTI DIME 2 est choisie nulle.

2.3 Type d'éléments

Les objets de type MAILLAGE constituent le support géométrique des éléments finis qui serviront à la modélisation des phénomènes physiques. Le type d'éléments de support de la géométrie est déclaré grâce au mot-clé ELEM.

Syntaxe :

OPTI ELEM MOT1 ;

où MOT1 est un mot pouvant prendre les valeurs listées dans le tableau 2.1.

Valeur de MOT1	Élément géométrique	Nombre de nœuds	Type d'interpolation
POI1	point	1	—
SEG2	segment	2	linéaire
SEG3		3	quadratique
TRI3	triangle	3	linéaire
TRI6		6	quadratique
QUA4	quadrangle	4	linéaire
QUA8		8	quadratique
CUB8	cube	8	linéaire
CU20		20	quadratique
PRI6	prisme	6	linéaire
PR15		15	quadratique
PYR5	pyramide	5	linéaire
PY13		13	quadratique
TET4	tétraèdre	4	linéaire
TE10		10	quadratique

TAB. 2.1 – Principaux types d'éléments de maillage disponibles

Nous avons listé ici les principaux types d'éléments géométriques mais d'autres éléments sont disponibles pour des formulations particulières (RAC2, RAC3, LIA3, LIA4, LIA6, LIA8, RAP3, LIP6, LIP8, TRI7, QUA9, CU27, TE15, PR21, MULT, POLY).

Notons que le type d'élément fini utilisé pour la modélisation des phénomènes physiques est déclaré lors de l'utilisation de l'opérateur MODEle. Le type des supports géométriques doit donc être en accord avec les éléments finis qui seront utilisés : par exemple, si l'on souhaite utiliser des éléments finis de type poutre, barre ou coque axisymétrique, les supports géométriques correspondants devront être des SEG2.

2.4 Densité (taille de maille)

La valeur de la densité correspond à la longueur de référence d'un élément de maillage. Ce paramètre peut être modifié grâce au mot-clé DENSite.

Syntaxe :

OPTI DENS FLOT1 ;

où FLOT1 est un réel représentant la densité courante. Cette option peut également être modifiée à l'aide de la directive DENSite en faisant : DENS FLOT1 ;



Chapitre 3

Définition de points

3.1 À partir des coordonnées

Un point est défini par ses coordonnées en utilisant l'opérateur POINT.

Syntaxe :

```
POIN1    = POIN FLOT1 (FLOT2) (FLOT3) ;  
POIN1    = FLOT1 FLOT2 (FLOT3) ;
```

où FLOT1, FLOT2 et FLOT3 sont respectivement la première, la seconde et la troisième coordonnée du point. La seconde syntaxe, qui ne fait pas appel à l'opérateur POINT, est bien entendu équivalente à la première mais n'est valable qu'en dimension 2 et 3. Pour chaque point créé, on lui associe la valeur de la densité courante en plus des coordonnées.

3.2 Par extraction sur des objets existant

Il est possible d'extraire, d'un maillage existant, les points vérifiant certaines propriétés en utilisant l'opérateur POINT. Deux syntaxes sont possibles :

- soit à partir d'un maillage, en isolant un point repéré par son numéro, le premier ou le dernier point, le point situé le plus près d'un point donné, les points situés sur une forme géométrique telle qu'une droite, un plan, un cylindre, un cône, une sphère ou un tore ;
- soit à partir d'un champ (CHPOINT ou MCHAML), en isolant les points où les valeurs du champ vérifient certaines relations (maximum, inférieur à, compris entre, etc. . .).

Syntaxe 1 :

```
MAIL2    = MAIL1 POIN | N1 ;  
          | 'INITIAL' ;  
          | 'FINAL' ;  
          | 'PROC'    POIN1 ;  
          | 'DROIT'   POIN1 POIN2 (FLOT1) ;  
          | 'PLAN'    POIN1 POIN2 POIN3 (FLOT1) ;  
          | 'CYLI'    AXE11 AXEJ2 POIN1 (FLOT1) ;  
          | 'CONE'    SOMM1 AXE11 POIN1 (FLOT1) ;  
          | 'SPHE'    CENTR1 POIN1 (FLOT1) ;  
          | 'TORE'    CENTR1 AXE11 CENTR2 POIN1 (FLOT1) ;
```

Syntaxe 2 :

```

MAIL2      = CHAMP1 POIN | 'MAXI' | ( 'AVEC' | )
              | 'MINI' | ( 'ABS' ) ( | LMOTS1 ) ;
              | | ( 'SANS' | )
              | 'SUPERIEUR' | | ;
              | 'EGSUPE' | | ;
              | 'EGALE' | FLOT1 | ;
              | 'EGINFE' | | ;
              | 'INFERIEUR' | | ;
              | 'DIFFERENT' | | ;
              | 'COMPRIS' FLOT1 FLOT2 | ;
    
```

Exemple :

```

MAIL2      = CHAMP1 POIN 'MAXI' 'ABS' ;
MAIL3      = MAIL1 POIN 'PROC' POIN1 ;
MAIL4      = MAIL1 POIN 'DROIT' POIN1 POIN2 (1.E-4) ;
    
```

MAIL2 est l'ensemble des points du maillage supportant le champ CHAMP1 où ce champ prend sa valeur maximale en valeur absolue.

MAIL3 est le point du maillage MAIL1 le plus proche du point POIN1.

MAIL4 est l'ensemble des points de MAIL1 appartenant à la droite définie par les deux points POIN1 et POIN2. Le critère de distance pour l'appartenance à cette droite, facultatif, est pris égal à 10^{-4} .

Notons que MAIL3 est un objet de type POINT alors que MAIL2 et MAIL4 sont des objets de type MAILLAGE dont les éléments sont de type POI1, même si ces maillage ne contiennent qu'un seul point.

Notons enfin que les procédures POINTCYL et POINTSPH permettent de définir un point par ses coordonnées respectivement cylindriques et sphériques.

Voir aussi : DANS, DEDANS, VISAVIS, ZIGZAG.



Chapitre 4

Définition de lignes

4.1 À partir de points

Une ligne permet de décrire plusieurs formes géométriques élémentaires telles que :

- un segment de droite (opérateur DROite)
- un arc de cercle (opérateur CERCcle, CER3, PARC)
- un arc de parabole (opérateur PARAbole)
- un arc de cubique (opérateur CUBP, CUBT)
- un arc d'ellipse (procédure ELLipse)
- un arc intersectant deux surfaces (opérateur INTERsection)
- un arc de courbe paramétré (opérateur COURbe, LIGNe)
- une ligne brisée (opérateur QUELconque)
- un congé de raccordement (opérateur CONGe)

Les opérateurs DROite et CERCcle peuvent être appelés par leurs alias respectifs D et C. Nous présentons ci dessous la syntaxe de l'opérateur DROite ainsi qu'un exemple d'utilisation. Il est possible de définir, soit le nombre d'éléments, soit la densité (taille) des éléments à chaque extrémité, soit ces deux paramètres en même temps. La syntaxe utilisée, pour définir le nombre et/ou la taille des éléments, est en partie commune à d'autres opérateurs pour réaliser d'autres types de lignes mais également des surfaces et des volumes.

Syntaxe :

```
DROI1 = DROI (ENTI1) POIN1 POIN2 ('DINI' FLOT1) ('DFIN' FLOT2) ;
```

Exemple :

```
DROI1 = D 12 POIN1 POIN2 ;  
DROI2 = D POIN1 POIN2 ;  
DROI3 = D POIN1 POIN2 'DINI' 0.1 'DFIN' 0.3 ;  
DROI4 = D ((-1)*7) POIN1 POIN2 'DINI' 0.4 'DFIN' 2. ;
```

Les quatre droites créées dans cet exemple joignent les points POIN1 et POIN2 de différentes manières :

- DROI1 possède 12 éléments de taille identique.
- DROI2 est divisée en éléments dont le nombre et la longueur sont calculés en fonction des densités affectées aux points POIN1 et POIN2.
- DROI3 est divisée en éléments dont le nombre et la longueur sont calculés en fonction des densités des extrémités. L'élément adjacent au point POIN1 aura une longueur égale à DINI (= 0.1) et l'élément adjacent au point POIN2 une longueur égale à DFIN (= 0.3).
- DROI4 est divisée en 7 éléments dont la taille est calculée en fonction des densités des extrémités (0.4 et 2).

Voir aussi : BOA, DEDouble, FUIT, HAUBAN, POUT2MAS, RACCord.

4.2 Par extraction sur des objets existant

Il est possible d'extraire certaines lignes d'un maillage existant en utilisant l'opérateur ELEMent dont la syntaxe est commune à tout type de maillage (lignes, surfaces, volumes). Son utilisation est décrite plus en détails dans la section 7.1.

Certains opérateurs sont dédiés à l'extraction de lignes de maillages existants :

- l'opérateur CONTour renvoie le contour d'un maillage 2D ;
- l'opérateur COTE renvoie un côté (ou les côtés) d'un maillage 2D ;
- l'opérateur ARETe renvoie les arêtes vives d'un maillage 3D.

Remarque : à l'exception de l'opérateur CONGe, toutes les lignes sont décrites par au moins un point (l'origine que l'on nomme POIN1) et une indication permettant d'obtenir l'extrémité (nommée POIN2). Si POIN1 est remplacé par un nom de ligne, c'est l'extrémité de cette ligne qui est l'origine de l'opérateur. Si POIN2 est remplacé par un nom de ligne, c'est l'origine de cette ligne qui est l'extrémité de l'opérateur.



Chapitre 5

Définition de surfaces

Le maillage d'une surface peut être obtenu par :

- découpage à partir des quatre cotés (opérateur DALLer) ;
- remplissage à l'intérieur d'un contour fermé (opérateur SURFace) ;
- opération géométrique sur une ligne (opérateurs TRANslation , ROTAtion) ;
- translation le long d'une ligne (opérateur GENEratrice) ;
- réglage entre deux lignes (opérateur REGLer) ;
- raccord entre deux lignes (opérateur COUTure).

5.1 À partir d'une ligne

Une surface peut être obtenue par TRANslation d'une ligne selon un vecteur ou ROTAtion d'une ligne selon un axe et un angle. Ces opérateurs fonctionnent de manière identiques à DROite pour contrôler la taille et/ou le nombre de mailles.

Syntaxe :

```
SURF1 = LIG1 TRAN (ENT11) ('DINI' FLOT1) ('DFIN' FLOT2) VEC1 ;  
SURF1 = LIG1 ROTA (ENT11) ANGL1 ('DINI' FLOT1) ('DFIN' FLOT2) ...  
... POIN1 (POIN2 si 3D) ;
```

Exemple :

```
SURF1 = LIG1 TRAN 23 VECT1 ;  
SURF2 = LIG1 TRAN ((-1)*6) 'DINI' 0.1 'DFIN' 0.05 VECT1 ;  
SURF3 = LIG1 ROTA 14 45. AXEI1 AXEJ2 ;
```

- SURF1 est la surface résultante de la translation de la ligne LIG1 selon le vecteur VECT1 (de type POINT) avec 23 éléments de même taille.
- SURF2 est la surface résultante de la translation de la ligne LIG1 selon le vecteur VECT1 (de type POINT) avec 6 éléments dont la taille est calculée en fonction des densités des extrémités (0.1 et 0.05).
- SURF3 est la surface résultante de la rotation de la ligne LIG1 selon l'axe défini par les points AXEI1 et AXEJ2 et d'angle 45 degrés en utilisant 14 éléments. En dimension 2, un seul point suffit à définir le centre de rotation.

Un maillage surfacique peut être réalisé à partir des lignes qui constituent son contour, par remplissage de l'intérieur du contour, à l'aide de l'opérateur SURFace. Les différentes options permettent de décrire une surface appuyée sur un objet géométrique élémentaire (plan, sphère, cylindre, cône ou tore) décrit par quelques points. Le contour fermé doit impérativement être située sur cet objet élémentaire.

Syntaxe :

```

SURF1      = SURF (CHPO1) LIG1 | 'PLANE' (FLOT1) ;
           | 'SPHERIQUE'   CENTR1 ;
           | 'CYLINDRIQUE' AXEI1 AXEJ1 ;
           | 'CONIQUE'     SOMM1 AXEJ1 ;
           | 'TORIQUE'     CENTR1 AXEJ1 CENTR2 ;
    
```

5.2 À partir de deux lignes

Une surface peut être obtenue à partir de deux lignes,

- soit par une translation de la première ligne le long de la seconde (opérateur GENEratrice) ;
- soit par le remplissage de l'espace entre ces deux lignes (opérateurs REGLer et COUture).

5.3 À partir de quatre lignes

L'opérateur DALLer permet de construire une surface à partir de quatre cotés. Chacun des cotés doit être maillé avec le même nombre de points, le même type d'éléments (SEG2 ou SEG3) et une orientation continue qui fournit l'orientation des éléments. De plus les quatre cotés doivent former un contour fermé. La syntaxe de l'opérateur DALLer, ainsi que ses options, sont similaires à l'opérateur SURFace.

5.4 Par extraction sur des objets existant

Il est possible d'extraire certaines surfaces d'un maillage existant en utilisant l'opérateur ELEMent dont la syntaxe est commune à tout type de maillage (lignes, surfaces, volumes). Son utilisation est décrite plus en détails dans la section 7.1.

Certains opérateurs sont dédiés à l'extraction de surfaces de maillages existants :

- l'opérateur ENVELOppe renvoie le maillage enveloppe d'un maillage 3D ;
- l'opérateur FACE renvoie la face (ou les faces) d'un maillage 3D.

Remarque : à l'exception de l'opérateur SURFace, toutes les surfaces sont décrites à partir d'au moins une ligne qui est le premier côté de cette surface et qui impose son orientation. Si cette ligne est remplacée par un nom de surface, c'est le troisième côté de cette surface qui fait fonction de la ligne qui va engendrer une nouvelle surface. Il faut bien entendu que cette surface soit constituée de quatre cotés c'est à dire qu'elle ait elle-même été générée par l'un des opérateurs DALLer, GENErer, REGLer, ROTation, TRANslation.

Exemple :

```

SURF2      = SURF1 TRAN VECT1 ;
    
```

SURF2 est la surface engendrée par la translation du troisième côté de SURF1 le long du vecteur VECT1.



Chapitre 6

Définition de volumes

6.1 Par remplissage d'une surface fermée

Le maillage d'un volume peut être obtenu par remplissage d'une surface fermée grâce à l'opérateur `VOLUME`.

Syntaxe :

```
VOLU1 = VOLU SURF1 ;
```

L'opérateur `PAVEr` permet de mailler l'intérieur d'un volume parallélépipédique à partir du maillage de chacune des six faces.

6.2 Par balayage d'une surface

Le maillage d'un volume peut être obtenu par balayage d'une surface également grâce à l'opérateur `VOLUME` et certains mots-clés.

Syntaxe :

```
VOLU1 = SURF1 VOLU (ENTI1) ('DINI' FLOT1) ('DFIN' FLOT2) ...  
      ... | 'TRAN' VEC1 ;  
          | 'ROTA' ANGL1 AXE11 AXEJ1 ;  
          | 'GENE' LIG1 ;  
          | SURF2 ;
```

- l'option `TRANslation` permet de d'engendrer un volume par translation (ou extrusion) d'une surface selon un vecteur ;
- l'option `ROTation` permet de générer un volume par rotation (ou révolution) d'une surface autour d'un axe défini par deux points et selon un angle ;
- de manière plus générale, l'option `GENEratrice` permet de générer un volume par balayage d'une surface selon une ligne ;
- si aucun mot-clé n'est précisé, le volume construit relie deux surface qui doivent être homéomorphes.

On notera que les mots-clés de l'opérateur `VOLUME` correspondent aux opérateurs homonymes dédiés à la génération de surfaces et que la syntaxe employée pour chacun de ces mots-clés est identique à celle de l'opérateur correspondant. Le nombre et/ou la taille des éléments générés est piloté par le nombre d'éléments `ENTI1` ou les densités initiales et finales `FLOT1` et `FLOT2`.

Remarque : à l'exception de l'opérateur VOLUme, tous les volumes sont décrits à partir d'au moins une surface qui est la première face de ce volume. Les faces ont une numérotation implicite accessible par l'opérateur FACE. Si cette surface est remplacée par un nom de volume, c'est la deuxième face de ce volume qui fait fonction de la surface qui va engendrer un nouveau volume. Il faut bien entendu que ce volume soit constitué de six faces c'est à dire qu'il ait lui-même été généré par l'un des opérateurs PAVEr, VOLU 'GENE', VOLU 'ROTA', VOLU 'TRAN'.

Exemple :

VOLU2 = VOLU1 VOLU TRAN VECT1 ;

VOLU2 est le le volume engendré par la translation de la deuxième face de VOLU1 le long de VECT1.



Chapitre 7

Manipulation de maillages

7.1 Extraction de parties

L'extraction d'une partie des éléments d'un maillage peut être réalisée par l'opérateur ELEMent qui fonctionne de manière analogue à l'opérateur POINt. Deux syntaxes sont possibles :

- soit à partir d'un maillage, en extrayant les éléments par leur numéro, leur couleur, les éléments contenant un point donné ou bien les éléments appuyés strictement ou largement un maillage donné ;
- soit à partir d'un champ (MCHAML), en isolant les éléments où les valeurs du champ vérifient certaines relations (maximum, inférieur à, compris entre, etc. . .).

Syntaxe 1 :

```
MAIL3      = MAIL1 ELEM | TYPE1 | LENT11 | ;
              | COUL1 | N1      | ;
              | 'CONTENANT' POIN1 ('TOUS') ;
              | 'APPUYE' ( | 'STRICTEMENT' | ) MAIL2 ('NOVERIF') ;
              |           ( | 'LARGEMENT'   | )
              | 'COMPRIS' POIN1 POIN2 ;
```

Syntaxe 2 :

```
MAIL1      = CHAMP1 ELEM | 'MAXI' |           ( 'AVEC'           )
              | 'MINI' | ('ABS') (           LMOTS1 ) ;
              |           |           ( 'SANS'           )
              | 'SUPERIEUR' |           | ;
              | 'EGSUPE'   |           | ;
              | 'EGALE'    | FLOT1   | ;
              | 'EGINFE'   |           | ;
              | 'INFERIEUR' |           | ;
              | 'DIFFERENT' |           | ;
              | 'COMPRIS'  FLOT1 FLOT2 | ;
```

Exemple :

```
MAIL2      = MAIL1 ELEM 'VERT' ;
MAIL3      = MAIL1 ELEM 'CU20' ;
MAIL4      = MAIL1 ELEM 'APPUYE' 'STRICTEMENT' MAIL2 ;
MAIL5      = MAIL1 ELEM 'APPUYE' 'LARGEMENT' MAIL2 ;

MAIL6      = CHE1 ELEM 'MAXI' ;
```

```
MAIL7 = CHE1 ELEM 'SUPERIEUR' 52. ;
MAIL8 = CHE1 ELEM 'DIFFERENT' 18. ;
MAIL9 = CHE1 ELEM 'COMPRIS' (-1.*2.) 140. ;
```

- MAIL2 est constitué des éléments de MAIL1 de couleur VERT ;
- MAIL3 est constitué des éléments de MAIL1 de type CU20 ;
- MAIL4 est constitué des éléments de MAIL1 appuyés *strictement* sur MAIL2, c'est-à-dire contenant la totalité des éléments de MAIL2 ;
- MAIL5 est constitué des éléments de MAIL1 appuyés *largement* sur MAIL2, c'est-à-dire contenant au moins un des nœuds des éléments de MAIL2 ;
- MAIL6 est constitué des éléments support du maximum des valeurs du champ CHE1 ;
- MAIL7 est constitué des éléments où la valeur du champ CHE1 est supérieure à 52 ;
- MAIL8 est constitué des éléments où la valeur du champ CHE1 est différente de 18 ;
- MAIL9 est constitué des éléments où la valeur du champ CHE1 est comprise largement entre -2 et 140.

On pourra également utiliser les opérateurs suivant pour extraire des éléments à partir de deux maillages :

- L'opérateur INCLUSION permet de déterminer les éléments d'un maillage se trouvant à l'intérieur d'un autre maillage. Plusieurs options sont possibles pour définir quand un élément est situé dans un autre maillage (un nœuds, tous les nœuds ou bien le barycentre).
- L'opérateur INTERsection détermine l'ensemble des éléments communs à deux maillages (les nœuds doivent être superposés).
- L'opérateur DIFFerence détermine la différence symétrique de deux maillages (union - intersection).

Voir aussi : CCON, COUPer, PARTition.

7.2 Transformations géométriques

Il est possible de créer un nouveau maillage en appliquant une transformation géométrique élémentaire à un maillage existant :

- Translation : opérateurs PLUS et MOINs (par rapport à un vecteur) ;
- Rotation : opérateur TOURner (par rapport à un axe et un angle) ;
- Symétrie : opérateur SYMEtrie (par rapport à un point, une droite ou un plan).
- Homothétie : opérateur HOMOthetie (par rapport à un point et un réel).
- Affinité : opérateur AFFInite (par rapport à une direction et un réel).
- Projection : opérateur PROJection. Les mots-clés permettent de définir une projection radiale (projection conique) ou selon un vecteur (projection cylindrique) ainsi que le support géométrique sur lequel on projette (plan , sphère, cylindre, cône, tore, droite ou cercle).

La directive DEPLacer permet de réaliser les mêmes transformations géométriques sans créer de nouveau maillage mais en transformant le maillage initial. Les mots-clés sont homonymes aux opérateurs et les syntaxes identiques.

7.3 Modifications

Les opérateurs et directives suivants permettent de modifier certaines parties ou certaines caractéristiques d'un maillage :

- L'opérateur CHANger permet de construire un maillage équivalent au maillage d'entrée mais formé d'éléments du type demandé. Il est également possible de convertir le maillage en éléments linéiques (option 'LIGNE') ou bien de remplacer chaque élément quadratique par un élément linéaire (option LINEAIRE) ou chaque élément linéaire par un élément quadratique (option QUADRATIQUE).



- L'opérateur INVERse permet de changer le sens de description d'une ligne (SEG2 ou SEG3) ou bien d'inverser l'orientation des éléments orientables d'une surface (TRI3, TRI6, QUA4, QUA8).
- L'opérateur ORIEnter permet de ré-orienter les éléments orientables d'un maillage.
- La directive ORDONner permet d'ordonner les numéros des éléments d'un maillage.
- La directive ELIMination permet de remplacer tous les nœuds d'un maillage, dont les distances entre eux sont inférieures à une certaine valeur, par un seul point. Le critère de distance des nœuds est à définir lors de l'appel à ELIMination ou, par défaut, pris comme le dixième de la valeur de densité courante. Cette directive est à utiliser avec précaution et parcimonie car la suppression de certains nœuds peut poser problème, notamment s'il existe dans un CHPOINT des valeurs différentes pour les nœuds doubles : il sera difficile de savoir quelle valeur garder.
- L'opérateur REGENERer permet de régénérer un maillage sur lequel on a appliqué la directive ELIMination en transformant les éléments où l'on a éliminé des nœuds doubles en éléments appropriés.
- La directive CONFondre permet de confondre deux points.
- La directive MODIfier permet la modification de maillages en interactif à l'aide d'une interface graphique. Elle permet notamment de déplacer et de nommer des points, de supprimer, d'ajouter ou de nommer des éléments d'un maillage.

Voir aussi : DANS, DEDANS, DEDU, RAFF, RAFT.

7.4 Contrôle et caractéristiques

Nous présentons quelques opérateurs et directives permettant de contrôler un maillage ainsi que d'obtenir certaines de ses caractéristiques :

- la directive TRACer permet de tracer un maillage sur l'unité graphique spécifiée dans OPTI TRAC ... ;
- l'opérateur NOEUd permet d'extraire un nœud d'un maillage à partir de son numéro, ou bien de connaître le numéro d'un nœud ;
- l'opérateur NBNO renvoie le nombre de nœuds d'un maillage ;
- l'opérateur NBEL renvoie nombre d'éléments d'un maillage ;
- l'opérateur BARYcentre renvoie le barycentre des points d'un maillage ;
- l'opérateur MESUre renvoie la mesure d'un maillage, c'est-à-dire, sa longueur s'il s'agit d'une ligne, sa surface s'il s'agit d'une surface et son volume s'il s'agit d'un volume ;
- l'opérateur COORdonnee renvoie les champs de coordonnées (type CHPOINT ou FLOTTANT) d'un maillage ou d'un point.

Voir aussi : INDI, REFE, SENS, VERM, VERSens.



Chapitre 8

Exemple de réalisation de maillage

8.1 Bague avec rainure extérieure

```
OPTI DIME 3 ELEM CUB8 ;
R      = 0.1 ;
L      = 1. ;
LIG1   = CERC 6 (R 0. 0.) (0. 0. 0.) (0. 0. R) ;
LIG2   = DROI 3 (L 0. 0.) (L 0. L) ;
LIG3   = DROI 3 LIG2 (0. 0. L) ;
SURF1  = REGL LIG1 LIG3 'DINI' 0.05 'DFIN' 0.2 ;
SURF2  = SURF1 ET (SURF1 SYME 'DROI' (0. 0. 0.) (1. 0. 0.)) ;
TRAC SURF2 ;
POIN6  = ((1.5*L) 0. 0.) ;
POIN7  = ((1.5*L) 0. 1.) ;
VOLU1  = SURF2 VOLU 40 ROTA 360. POIN6 POIN7 ;
ELIM 1.E-10 VOLU1 ;
VOLU2  = REGE VOLU1 ;
TRAC FACE VOLU2 ;
FIN ;
```




FIG. 8.1 – Section plane et maillage volumique par révolution d'une bague avec rainure extérieure (éléments 3D massifs).

8.2 Casserole simple en éléments coques

```
OPTI DIME 3 ELEM QUA4 ;
R1      = 1.5 ;
R2      = 0.2 ;
H1      = 1.5 ;
L1      = 1.8 ;
LIG1    = DROI 8 (0. 0. 0.) ((R1-R2) 0. 0.) ;
LIG2    = DROI 10 (R1 0. R2) (R1 0. (R2+H1)) ;
A LIG3 B = LIG1 CONG 5 R2 LIG2 ;
LIGTOT  = A ET LIG3 ET B ;
SURF1   = LIGTOT ROTA 60 360. (0. 0. 0.) (0. 0. 1.) ;
ELIM 1.E-10 SURF1 ;
SURF2   = REGE SURF1 ;
POIN1   = SURF2 POIN PROC (R1 R1 (R2+(0.8*H1))) ;
ZPOIN1  = POIN1 COOR 3 ;
SURF2LIG = CHAN 'LIGNE' SURF2 ;
CHZ     = SURF2LIG COOR 3 ;
MPOIN1  = CHZ POIN 'EGAL' ZPOIN1 ;
CHY     = MPOIN1 COOR 2 ;
MPOIN2  = CHY POIN 'COMPRIS' (-0.2*R1) (0.2*R1) ;
CHX     = MPOIN2 COOR 1 ;
MPOIN3  = CHX POIN 'SUPERIEUR' 0. ;
LIG4    = SURF2LIG ELEM 'APPUYE' 'STRICTEMENT' MPOIN3 ;
SURF3   = LIG4 TRAN 10 (L1*(1. 0. 0.3)/(NORM (1. 0. 0.3))) ;
SURFTOT = SURF2 ET SURF3 ;
TRAC FACE SURFTOT ;
FIN ;
```




FIG. 8.2 – Maillage simple d'une casserole par révolution (éléments coques).



Chapitre 9

Aide mémoire

9.1 Opérateurs généraux

BARY	Calcule le barycentre des points d'un maillage (moyenne arithmétique de l'ensemble des nœuds).
@CDG	Calcule les coordonnées du centre de gravité de la géométrie contenue dans un modèle.
COOR	Renvoie les coordonnées d'un objet de type POINT, MAILLAGE, CHPOINT ou MCHAML.
COUL	Duplique un objet en lui attribuant une couleur choisie.
DENS	Opérateur de base pour définir la densité courante.
MESU	Calcule la mesure du maillage.
@MESU	Calcule la mesure d'un maillage.
NBEL	Renvoie le nombre d'éléments contenus dans un maillage.
NBNO	Renvoie le nombre de nœuds contenus dans un maillage.
REFE	Cet opérateur a deux fonctions : <ol style="list-style-type: none">1. Liste les objets maillages inclus dans un autre.2. Indiquer si un objet maillage est inclus dans un autre.
SENS	Cet opérateur a deux fonctions : <ol style="list-style-type: none">1. Calcule la différence de deux MCHAMLS puis fait la moyenne arithmétique de la valeur du champ aux points de Gauss pour avoir la valeur moyenne du champ sur chaque sous-zone.2. Détermine le sens de parcours d'un ou plusieurs contours orientés fermés en dimension 2.
VERM	Vérifie un maillage constitué d'éléments massifs.

VERS Produit une erreur si deux éléments jointifs d'un maillage sont orientés en sens opposés.

9.2 Définition de points

EXTC Renvoie les points centre ou face d'un objet maillage de type FACEL.

NOEU Permet d'identifier un nœud dans un maillage ou de connaître le numéro actuel d'un nœud d'après son nom.

POIN Permet de créer des points et de récupérer un ou plusieurs points d'un maillage.

POINTCYL Définit un point par ses coordonnées cylindriques.

POINTSPH Définit un point par ses coordonnées sphériques.

9.3 Définition de lignes

ARET Construit un maillage constitué d'éléments SEG2 représentant les arêtes vives d'un maillage tridimensionnel.

BOA Construit le maillage d'une ligne de tuyauterie de façon interactive.

CER3 Construit un arc de cercle défini par trois points.

CERC Construit un arc de cercle.

CONG Construit un congé de raccordement (circulaire) entre deux lignes.

CONT Construit le contour d'un maillage.

COTE Rend la ligne située sur un bord d'un maillage de type quadrilatère.

@COUPLE Crée un couple, c'est-à-dire un maillage de segments à deux nœuds à partir de deux EVOLUTIONs.

COUR Crée une courbe polynomiale.

COUT Construit une surface qui relie deux lignes ou une ligne et un point à l'aide de triangles.

CUBP Construit la ligne cubique passant par quatre points.

CUBT Construit un arc de cubique passant par deux points et avec des tangentes.

DEDO Dédoublé les nœuds d'une ligne.



9.4. DÉFINITION DE SURFACES

DROI	Construit le maillage d'une droite.
HAUBAN	Construit le maillage d'un câble accroché à ses extrémités et soumis à son poids propre ainsi que le modèle associé et le champ de tensions dans le câble.
INVE	Cet opérateur a trois fonctions : <ol style="list-style-type: none">1. Change le sens de description des lignes d'un maillage, dont les éléments sont de type SEG2/SEG3.2. Inverse l'orientation des éléments « orientables » d'un maillage (éléments de type TRI3/QUA4/TRI6/QUA8).3. Calcule l'inverse d'un CHPOINT au sens de la multiplication *.
LIGN	Maillage d'un cercle à partir d'un point et d'un angle ou d'une droite à partir d'un point et d'un vecteur.
PARA	Construit un arc de parabole joignant deux points en spécifiant le point d'intersection des tangentes en ces points
PARC	Construit un arc de cercle en l'approchant par une suite de paraboles.
QUEL	Construit une ligne brisée.
ZIGZAG	Construit une ligne définie par une succession de parties droites et arrondies.

9.4 Définition de surfaces

COUP	Construit la coupe 2D d'un maillage 3D selon un plan.
DALL	Maillage de surfaces entre quatre (ou trois) lignes.
ENVE	Cet opérateur a deux fonctions : <ol style="list-style-type: none">1. Construit l'enveloppe d'un maillage volumique.2. Construit le spectre enveloppe d'une série de spectres d'oscillateurs.
FACE	Rend une face d'un maillage volumique.
GENE	Construit la surface engendrée par la translation d'une ligne parallèlement à une autre.
MAILSTRU	Construit un maillage structuré à partir d'un contour.
REGL	Construit la surface réglée s'appuyant sur deux lignes.
ROTA	Construit la surface engendrée par la rotation d'une ligne, selon un angle donné autour d'un point en 2D ou d'un axe en 3D.

SURF	Construit le maillage de l'intérieur du contour défini par un ensemble de lignes fermées. Permet également de construire une surface paramétrée.
TRAN	Construit la surface engendrée par la translation d'une ligne suivant un vecteur donné.

9.5 Définition de volumes

COQ2MAS	Construit un maillage volumique à partir d'un modèle de coque pour permettre la vérification de ses dimensions et de ses orientations.
CREER_3D	Visualisation 3D de certains résultats 2D axis.
FOUR2TRI	Génère un maillage 3D à partir d'un modèle 2D Fourier.
PAVE	Maille avec des cubes l'intérieur d'un volume parallélépipédique dont les six faces sont précisées.
POUT2MAS	Construit un maillage massif à partir d'un modèle de poutre pour permettre la vérification de ses dimensions et de ses orientations.
VOLU	Maillage de volume à partir d'une surface.

9.6 Manipulation de maillages

AFFI	Réalise une affinité géométrique.
CCON	Sépare les composantes connexes d'un maillage et les met dans un objet table indicé par des entiers.
CONF	Confond deux points.
COUPLER	Construit un maillage décalé d'une demi-épaisseur et crée des éléments raccord.
DEDU	Déduction d'un maillage à partir de la transformation d'un certain nombre de nœuds.
@DEDUIRE	Construit un nouveau maillage en changeant les nœuds maîtres d'un maillage donné.
DEPL	Déplacement d'un maillage.
DIFF	Construit la différence symétrique (union - intersection) de deux maillages.



9.6. MANIPULATION DE MAILLAGES

ELEM	Cet opérateur a deux fonctions : <ol style="list-style-type: none">1. Extrait une partie des éléments d'un maillage.2. Renvoie les types ou les couleurs des éléments d'un maillage.
ELIM	Élimination de nœuds doubles.
FUIT	Permet de scinder un contour orienté fermé en deux autres contours orientés fermés et de créer le segment de fuite (type SEG2) qui les relie.
HOMO	Construit un objet par homothétie.
IJET	Calcule, par une méthode analytique exacte, l'intersection entre des segments et les facettes triangulaires d'un maillage.
INCL	Revoie les éléments d'un maillage à l'intérieur d'un contour, d'une surface ou d'un volume.
INDI	Fournit des indicateurs permettant d'apprécier la qualité d'un maillage.
INTE	Cet opérateur a deux fonctions : <ol style="list-style-type: none">1. Construit l'intersection de deux maillages, c'est à dire l'ensemble des éléments appartenant aux deux maillages.2. Construit l'arc de courbe, intersection de deux surfaces, compris entre deux points.
ITRC	Calcule, par une méthode analytique exacte, l'intersection entre des segments et les facettes triangulaires d'un maillage.
ORDO	Ordonne les objets de type LISTREEL, LISTENTI, MAILLAGE ou EVOLUTION.
ORIE	Construit un maillage identique au maillage initial, mais dont tous les éléments orientables sont orientés dans le sens trigonométrique en 2D, ou en fonction de leur direction par rapport à un vecteur ou à un point en 3D.
PART	Construit une partition d'un maillage en tentant de minimiser le nombre de points sur les frontières et le déséquilibre entre les parties.
PROJ	Projection d'un maillage sur un lieu géométrique.
RAFF	Raffine un maillage en respectant un champ de densité.
RAFT	Raffine un maillage triangulaire en respectant une carte de taille donnée par des valeurs aux nœuds.
REGE	Régénère un maillage sur lequel on a appliqué la directive ELIM en transformant les éléments où l'on a éliminé des nœuds doubles en éléments appropriés.

SYME	Construit le maillage et le CHPOINT résultants de la symétrie d'un maillage et d'un CHPOINT éventuellement associé.
TOUR	Crée un objet résultant de la rotation autour d'un point ou d'un axe du support géométrique d'un objet, et éventuellement de ses composantes.

9.7 Autres opérations

ANNOIMP	Construit le maillage d'anneau imparfait.
@CARENE	Crée une carène (maillage tridimensionnel formé d'éléments de type QUA4) à partir de couples (formés d'éléments de type SEG2).
CONTSEG3	Création d'une rigidité et de deux maillages en vue de contacts unilatéraux.
@COUTOR1	Calcule la courbure et la torsion d'un segment de ligne.
@COUTOR2	Calcule la courbure et la torsion d'une ligne sur chacun de ses éléments.
DANS	Teste l'inclusion : <ul style="list-style-type: none"> – d'une liste (LISTENTI, LISTREEL) dans une autre ; – d'un point dans l'ensemble des nœuds d'un maillage.
DEDANS	Teste si un point est situé à l'intérieur d'un contour orienté fermé.
@FIS_3DS	Crée un bloc fissure 3D massif en utilisant des éléments CU20 et PR15. La fissure est supposée elliptique.
GENJ	Génère le maillage d'éléments de joint susceptibles de lier les contours intérieurs d'un maillage.
@HELICE	Crée un maillage engendré par une transformation hélicoïdale.
IMPF	Crée un maillage d'élément de frottement à partir d'un maillage d'éléments de contacts ou d'un maillage de SEG2 pour le frottement de câbles.
IMPO	Cet opérateur a deux fonctions : <ol style="list-style-type: none"> 1. Création d'un maillage en vue de contacts unilatéraux. 2. Permet d'imposer une condition de contact entre deux lignes (de type SEG2) en 2D (mode PLAN ou AXI) ou deux surfaces (de type TRI3) en 3D.
LIAI	Construit l'ensemble des éléments de liaison entre deux objets surfaciques.
MPRO	Construit un maillage de la matrice triangulaire inférieure apparaissant dans la factorisation d'une matrice.
RACC	Crée un élément de raccord ordinaire ou poreux.



9.7. AUTRES OPÉRATIONS

@RAYO	Crée un maillage rayonnant symétrique autour de la pointe d'une fissure.
@REPERE	Construit un repère (x, y, z) .
@TOLE2	Crée un maillage massif de CUB8 à partir d'un maillage de carènes (maillage surfacique tridimensionnel QUA4).
@TOLE3	Crée un maillage massif de CUB8 à partir de deux couples de carènes ayant le même nombre de points (maillage 3D de SEG2).
VISA	Crée la liste des nœuds de deux maillages localisés au même endroit.



Index

A		DENS	10, 25
AFFI	20, 28	Densité	10
ANNOIMP	30	DEPL	20, 28
ARET	14, 26	Dessiner	21
B		DIFF	20, 28
Balayage	17	Dimension	9
BARY	21, 25	DROI	13, 27
BOA	26	E	
C		ELEM	19, 29
@CARENE	30	ELIM	7, 21, 29
CCON	28	ELLI	13
@CDG	25	ENVE	16, 27
CER3	13, 26	ET	7
CERC	13, 26	EXTC	26
CHAN	20	Extrusion	17
CONF	21, 28	F	
CONG	13, 26	FACE	16, 27
CONT	14, 26	@FIS_3DS	30
CONTSEG3	30	FOUR2TRI	28
COOR	21, 25	FUIT	29
COQ2MAS	28	G	
COTE	14, 26	GENE	15, 27
COUL	25	GENJ	30
Couleur	9	H	
COUP	27	HAUBAN	27
@COUPLE	26	@HELICE	30
COUPLER	28	HOMO	20, 29
COUR	13, 26	I	
COUT	15, 16, 26	IJET	29
@COUTOR1	30	IMPF	30
@COUTOR2	30	IMPO	30
CREER_3D	28	INCL	20, 29
CUBP	13, 26	INDI	29
CUBT	13, 26	INTE	13, 20, 29
D		INVE	21, 27
DALL	15, 16, 27	ITRC	29
DANS	30	L	
DEDANS	30	LIAI	30
DEDO	27	LIGN	13, 27
DEDU	28		
@DEDUIRE	28		

Lignes	13	Surfaces.....	15
		SYME.....	20, 30
M			
MAILSTRU.....	27	T	
MESU.....	21, 25	Taille de maille	10
@MESU.....	25	@TOLE2.....	31
MODI.....	21	@TOLE3.....	31
MOIN.....	20	TOUR.....	20, 30
MPRO.....	30	TRAC.....	21
		Tracer.....	21
N			
NBEL.....	21, 25	TRAN.....	15, 28
NBNO.....	21, 25	Translation.....	15, 17
NOEU.....	21, 26	Type d'éléments	9
O			
OPTI.....	9	V	
Options.....	9	VERM.....	26
ORDO.....	21, 29	VERS.....	26
ORIE.....	21, 29	VISA.....	31
		VOLU.....	17, 28
		Volumes	17
P			
PARA.....	13, 27	Z	
PARC.....	13, 27	ZIGZAG.....	27
PART.....	29		
PAVE.....	17, 28		
PLUS.....	20		
POIN.....	11, 26		
POINTCYL.....	12, 26		
Points.....	11		
POINTSPH.....	12, 26		
POUT2MAS.....	28		
PROJ.....	20, 29		
Q			
QUEL.....	13, 27		
R			
RACC.....	31		
RAFF.....	29		
RAFT.....	29		
@RAYO.....	31		
REFE.....	25		
REGE.....	21, 29		
REGL.....	15, 16, 27		
@REPERE.....	31		
Révolution.....	17		
ROTA.....	15, 27		
Rotation.....	15, 17		
S			
SENS.....	25		
SURF.....	15, 28		